
Humpty Dumpty's Egg Drop

The 40th game in 1K hires on the ZX81. This time again with a famous person.

Humpty Dumpty made his entrance to the ZX81.  Don't drop Humpty Dumpty....

Please let him land safely on the next platform. He is fragile. 

; Humpty Dumpty's Egg Drop

; Controls 

; Start = Newline

; Left  = 1-5

; Right = 6-0                                                   

                                                                

? * TORNADO *                                                   

                                                                

height     EQU  168-64                                          

                                                                

           ORG  #4009             ;#4009                        

           DUMP 49161                                           

                                                                

basic      LD   B,0               ; line number but also code   

           JR   init0             ; length                      

                                                                

           DEFB 236,212,28        ; GOTO USR 0                  

           DEFB 126                                             

           DEFB 143,0,18          ; shortest FP-number of #4009 

                                                                

nline      DEFW last              ; NEEDED for loading          

chadd      DEFW last-1                                          

xptr       DEFW 0                                               

stkbot     DEFW last                                            

stkend     DEFW last                                            



berg       DEFB 0                                               

mem        DEFW 0                 ; not needed without fp       

           DEFB 128                                             

                                                                

init1      JP   init                                            

                                                                

lastk      DEFB 255,255,255       ; above reusable after loading

margin     DEFB 55                                              

                                                                

nxtlin     DEFW basic             ; NEEDED for loading only     

                                                                

init0      XOR  A                 ; start 48K bugfix            

           DEFB 254               ; CP n, skip flagx            

flagx      DEFB 0                                               

           JR   init1             ; continue                    

                                                                

taddr      DEFW 3213              ; used by ZX                  

                                                                

           DEFB 0                                               

           DEFB 0                                               

                                                                

frames     DEFW 65535             ; unusable block              

coords     DEFB 0,0                                             

prcc       DEFB 188                                             

sposn      DEFB 33,24                                           

cdflag     DEFB 64                ; end unusable block          

                                                                

           DEFW 0,0               ; filler                      

                                                                

; humpty dumpty graphics                                        

; 1 start left leg                                              

; 2 start right leg                                             

; 3 second start left leg                                       

; 4 second start right leg                                      

; 5 first drop miss                                             

; 6 second drop miss                                            

                                                                

hdudg1     DEFB 0,60,126,153,221,153                            

           DEFB 255,219,102,60,38,96,0                          

hdudg2     DEFB 0,60,126,153,187,153                            

           DEFB 255,219,102,60,100,6                            

           DEFB 0,0,0,0,0,0,0                                   

hdudg3     DEFB 0,52,110,217,189,121                            

           DEFB 255,219,102,60,38,96,0                          

hdudg4     DEFB 0,52,110,217,187,121                            

           DEFB 255,219,102,60,100,6,0                          

                                                                

hdudg5     DEFB 102,36,60,126,195,255                           

           DEFB 175,222,173,251,118,44                          

hdudg6     DEFB 203,163,202,170,0,231,60                        

           DEFB 126,190,221,235,118                             

                                                                

hr         LD   HL,lowres+#8000   ; the lowres display          

           LD   BC,#209                                         

           LD   A,#1E                                           

           LD   I,A                                             

           LD   A,#FB                                           

           CALL #2B5                                            

                                                                

           LD   B,7               ; sync hires display          

hr00       DJNZ hr00                                            

           LD   A,(HL)                                          

                                                                

           LD   HL,scrlines       ; the screen with platforms   



           LD   A,H                                             

           LD   D,H                                             

           LD   I,A               ; set high byte of hr-screen  

           EXX                                                  

           LD   HL,scrudg         ; the Humpty-UDG screen       

           LD   BC,#B10F          ; 176 lines +1                

                                                                

low        DEC  B                 ; on RET here NEVER to zero   

                                                                

lineloop   LD   A,B               ; check Humpty on screen      

           CP   (HL)                                            

           JR   NZ,testscr        ; test platform-display       

                                                                

doudg      INC  L                 ; display Humpty-udg          

           LD   E,(HL)            ; fetch position of item      

           INC  HL                                              

           LDI                    ; copy item to LBUF           

           LD   C,E               ; save "position" for erase   

           LD   A,0               ; XOR A is too fast           

           LD   R,A               ; set lowbyte                 

           JP   #C000             ; display the item            

                                                                

testscr    EXX                                                  

           CP   (HL)              ; check platform              

           JR   NZ,doempty        ; no platform, empty line     

                                                                

doscr      INC  HL                ; display platformline        

           LD   A,L                                             

           LD   L,scrline2*256/256;pointer second line          

           EXX                                                  

           LD   R,A                                             

           JP   #C000             ; show platform               

                                                                

doempty    LD   A,7                                             

delay      DEC  A                 ; timefill an empty line      

           JR   NZ,delay                                        

                                                                

           EXX                    ; back to maintests           

                                                                

           LD   E,C               ; erase last printed item     

           DEC  DE                ; from LDI 1 too far          

           LD   (DE),A            ; erase for next item         

           LD   (DE),A            ; timing                      

           INC  DE                ; timing                      

                                                                

           DJNZ lineloop          ; LAST LINE EMPTY, only exit  

                                                                

           CALL #292              ; back from intrupt           

           CALL #220                                            

           LD   IX,hr                                           

           JP   #2A4                                            

                                                                

doline     LD   A,(HL)            ; get X on line               

           AND  #1F               ; 0 - 31                      

           DEFB #CA               ; JP C never true             

undo       ADD  A,14                                            

           SUB  7                 ; 31-3-3 line=28 platform=3   

           JR   C,undo            ; original less than 6        

           INC  A                 ; at least 1                  

           LD   (DE),A            ; save X-position             

           RET                                                  

                                                                

cls        LD   BC,#3C00          ; clear all previous platforms

           LD   HL,scrlines                                     



cls1       LD   (HL),C                                          

           INC  HL                                              

           DJNZ cls1                                            

           RET                                                  

                                                                

eogtest    JR   NZ,nextloop       ; end of game jump            

                                                                

           LD   BC,5              ; Game over, test hiscore     

           LD   DE,hiscore-1                                    

           LD   HL,score-1                                      

fihi       INC  HL                                              

           INC  DE                                              

           DEC  C                                               

           JR   Z,start                                         

           LD   A,(DE)                                          

           CP   (HL)                                            

           JR   Z,fihi                                          

           JR   NC,start                                        

           LDIR                                                 

                                                                

start      LD   A,%10111111       ; game over, wait for         

           IN   A,(254)                                         

           RRA                    ; newline                     

           JR   C,start                                         

                                                                

loadstart  LD   HL,0                                            

           LD   (strom+1),HL      ; reset ROM start             

                                                                

           LD   HL,#1C1C          ; reset score                 

           LD   (score),HL                                      

           LD   (score+2),HL                                    

                                                                

nextloop   CALL cls               ; dropped Humpty, restart     

                                                                

           LD   HL,init0                                        

           LD   (HL),1            ; set dx line 1               

           INC  HL                                              

           LD   (HL),255          ; opposite move line 2        

                                                                

strom      LD   HL,0              ; startline in ROM, increases 

           LD   DE,xlines+1                                     

           CALL doline            ; set first line              

                                                                

           LD   A,(DE)                                          

           LD   (yplay+2),A       ; Humpty Dumpty platform X 

                                                                

           INC  DE                                              

           INC  HL                                              

                                                                

botlinein  CALL doline            ; set second or new line      

                                                                

           LD   A,height                                        

           LD   (platformy+1),A                                 

           ADD  A,12                                            

           LD   (yplay+1),A       ; Humpty Dumpty platform Y 

                                                                

; built screen                                                  

platformy  LD   E,height                                        

           LD   D,2                                             

xlines     LD   BC,#0202                                        

           LD   A,B               ; swap needed                 

           LD   B,C                                             

           LD   C,A                                             

           LD   HL,scrlines                                     



xline2     LD   (HL),E                                          

           XOR  A                                               

clline     INC  HL                                              

           LD   (HL),A            ; clear before platform       

           DJNZ clline                                          

           LD   B,C               ; save second x position      

           DEC  A                 ; now 255                     

           LD   (HL),A                                          

           INC  HL                                              

           LD   (HL),A                                          

           INC  HL                                              

           LD   (HL),A            ; 3 positions platform        

           INC  HL                                              

           INC  A                                               

           LD   (HL),A            ; erase 3x after platform     

           INC  HL                                              

           LD   (HL),A                                          

           INC  HL                                              

           LD   (HL),A                                          

           LD   L,scrline2*256/256                              

           LD   A,E               ; second line data            

           SUB  64                                              

           LD   E,A                                             

           DEC  D                                               

           JR   NZ,xline2         ; do second line              

                                                                

           LD   BC,(udg1+1)       ; get current Humpty pointer  

yplay      LD   DE,#600           ; y and x of player           

           LD   HL,scrudg         ; the screen part of Humpty   

                                                                

sus        PUSH HL                ; save screenposition         

           LD   HL,scrlines                                     

           LD   A,E               ; on same line 1st platform?  

           CP   (HL)                                            

           JR   Z,onscreen        ; if so set on that line      

           LD   L,scrline2*256/256                              

           CP   (HL)              ; second platform?            

           JR   Z,onscreen        ; do the same                 

           POP  HL                ; get screenposition          

                                                                

setscrudg  LD   (HL),E            ; Y of udg Humpty             

           INC  HL                                              

           LD   (HL),D            ; X of humpty                 

           INC  HL                                              

           LD   A,(BC)                                          

           LD   (HL),A            ; the UDG data of Humpty      

           INC  HL                                              

           JR   nudgline                                        

                                                                

onscreen   LD   A,L                                             

           ADD  A,D                                             

           LD   L,A                                             

           INC  HL                ; calculate screen X          

           LD   A,(BC)                                          

           LD   (HL),A            ; write directly              

           POP  HL                ; retrieve screenposition     

                                                                

nudgline   LD   (HL),E            ; impossible next line        

           DEC  E                 ; next udg line               

           INC  BC                                              

           LD   A,(BC)                                          

           OR   A                 ; test end of Humpty UDG-data 

           JR   NZ,sus            ; set UDG on screen           

                                                                



udg1       LD   HL,hdudg1                                       

           LD   A,L                                             

           XOR  #0D                                             

           LD   L,A                                             

           LD   (udg1+1),HL       ; swap feet and eyes          

                                                                

           LD   HL,frames         ; delay for playable game     

           LD   A,(HL)                                          

           SUB  10                                              

wfr        CP   (HL)                                            

           JR   NZ,wfr                                          

                                                                

           LD   HL,xlines+2       ; x storage                   

           LD   DE,init0          ; dx storage                  

           LD   A,2                                             

           LD   (noplat+1),A      ; not on platform count       

           LD   B,A                                             

domove     LD   A,(DE)            ; get dx                      

           ADD  A,(HL)            ; add xpos                    

           LD   C,A               ; save new                    

           SUB  2                                               

           CP   25                ; <0 or >25 is swap           

           JR   C,setnewx                                       

           LD   A,(DE)            ; swap direction              

           NEG                                                  

           LD   (DE),A                                          

           JR   domove            ; move again other way around 

                                                                

setnewx    LD   A,(yplay+2)       ; get player x                

           SUB  (HL)              ; test against old x platform 

           LD   (HL),C            ; set new x                   

           INC  A                                               

           CP   3                                               

           JP   NC,noplat         ; not on platform for x       

                                                                

           LD   A,(platformy+1)   ; X on platform, now test Y   

           DEC  B                                               

           JR   Z,plat2           ; test top/bottom platform    

           SUB  64                ; calculate lower platform    

plat2      ADD  A,12                                            

           INC  B                 ; undo djnz                   

           LD   C,A                                             

           LD   A,(yplay+1)       ; player y                    

           CP   C                                               

           JR   NZ,noplat         ; no match, not on platform   

                                                                

; movement only allowed on platforms                            

           LD   C,0               ; total sum dx                

           LD   A,(lastk)                                       

           CP   %11110111         ; 1-5                         

           JR   NZ,right          ; test left                   

           DEC  C                 ; move left                   

right      CP   %11101111         ; 6-0                         

           JR   NZ,platdir        ; test right                  

           INC  C                 ; move right                  

                                                                

platdir    LD   A,(yplay+2)       ; get x                       

           EX   DE,HL                                           

           ADD  A,(HL)            ; add platform movement       

           ADD  A,C               ; add dx                      

           EX   DE,HL                                           

           LD   (yplay+2),A       ; save new x                  

           LD   A,(yplay+1)                                     

           CP   height-64+12      ; test landed on bottom       



           JR   NZ,onplat         ; continue if not             

                                                                

; now on bottom platform, lift up                               

lift       LD   HL,frames                                       

           LD   A,(HL)                                          

wflift     CP   (HL)                                            

           JR   Z,wflift                                        

           LD   B,12              ; 12 lines of Humpty graphic  

           LD   HL,scrudg                                       

upudg      INC  (HL)              ; 2 lines up                  

           INC  (HL)                                            

           INC  HL                                              

           INC  HL                                              

           INC  HL                                              

           DJNZ upudg             ; do full Humpty up           

           LD   A,(scrlines)      ; get Y platform 1            

           ADD  A,2                                             

           LD   (scrlines),A      ; add 2                       

           SUB  64                                              

           LD   (scrline2),A      ; also second platform up     

           CP   height            ; repeat until bottom on top  

           JR   NZ,lift                                         

                                                                

           LD   HL,score+4        ; now add a point             

           DEFB 1                 ; hide “ten”                               

ten        LD   (HL),28                                         

           DEC  HL                                              

           INC  (HL)              ; add a point                              

           LD   A,(HL)                                          

           CP   38                ; > “9” ?                              

           JR   Z,ten                                           

                                                                

           CALL cls               ; clear platforms             

                                                                

           LD   HL,(strom+1)      ; STart ROM                             

           INC  HL                                              

           LD   (strom+1),HL      ; next ROM-pointer            

           PUSH HL                                              

                                                                

           LD   HL,init0+1        ; data bottom dx to top       

           CALL swap                                            

                                                                

           LD   HL,xlines+2       ; data bottom x to top        

           CALL swap                                            

           EX   DE,HL                                           

           POP  HL                                              

           JP   botlinein         ; add a new line              

                                                                

swap       LD   C,(HL)                                          

           DEC  HL                                              

           LD   A,(HL)                                          

           LD   (HL),C                                          

           INC  HL                                              

           LD   (HL),A                                          

           RET                                                  

                                                                

noplat     LD   A,0               ; count the times on platform 

           DEC  A                                               

           LD   (noplat+1),A                                    

                                                                

onplat     DEC  HL                                              

           INC  DE                                              

                                                                

           DEC  B                                               



           JP   NZ,domove                                       

                                                                

droptest   LD   A,(noplat+1)                                    

           OR   A                                               

           JR   NZ,nodrop         ; not zero, then on a platform

                                                                

           LD   A,(yplay+2)       ; test remains of Humpty      

           ADD  A,scrline2*256/256                              

           INC  A                                               

           LD   L,A                                             

           LD   H,scrlines/256                                  

           LD   A,(HL)                                          

           INC  A                                               

           JR   Z,skiperase       ; do not erase platform       

           LD   (HL),0            ; erase possible remains      

skiperase  LD   A,(yplay+1)                                     

           SUB  8                 ; drop 8 lines                

           CP   14                                              

           LD   (yplay+1),A                                     

           JR   C,droppedhd       ; test bottom hit             

                                                                

nodrop     JP   platformy                                       

                                                                

droppedhd  LD   HL,hdudg6         ; alter dropped Humpty UDG    

           LD   A,L                                             

           XOR  #FC                                             

           LD   L,A                                             

           LD   (droppedhd+1),HL                                

           LD   DE,scrudg+2       ; copy dropped UDG to screen  

           LD   BC,#0CFF                                        

copy       LDI                                                  

           INC  DE                                              

           INC  DE                                              

           DJNZ copy                                            

                                                                

           LD   HL,frames         ; show dropped Humpty         

           LD   A,(HL)            ; for some time               

           SUB  80                                              

wdead      CP   (HL)                                            

           JR   NZ,wdead                                        

                                                                

liveloop   LD   A,hdudg1*256/256  ; alter to next humpty        

           XOR  32                                              

           LD   (udg1+1),A                                      

           LD   (liveloop+1),A                                  

           CP   hdudg1*256/256    ; second drop resets          

           JP   eogtest                                         

                                                                

space      EQU  #4300-$                                         

           DEFS space                                           

                                                                

           DEFB 0                                               

lbuf       DEFW 0,0,0,0,0,0,0                                   

           DEFW 0,0,0,0,0,0,0                                   

                                                                

; 3 lines for platforms, y,nl,28 bytes                          

scrlines   JP   low               ; start of platformlines      

                                                                

init       LD   IX,hr             ; Hires mode                  

           LD   SP,#4300          ; SP in game, not at end      

           EX   AF,AF'            ; delay intrupt               

                                                                

           LD   HL,lbuf           ; linebuffer over sysvar      

           LD   DE,#4000                                        



           LD   C,31              ; this copy needed            

           LDIR                   ; before 48K repair           

                                                                

           LD   L,C               ; 0                           

           LD   H,D               ; HL = #4000                  

           LD   E,L                                             

           LD   D,#C0             ; DE = #c000                  

           LD   B,4               ; BC = #400                   

           LDIR                   ; repair 48K bug              

           NOP                                                  

                                                                

scrline2   JP   loadstart                                       

           DEFS 30                                              

                                                                

; final byte as fake marker                                     

                                                                

x          EQU  101                                             

                                                                

           DEFB 118                                             

lowres     DEFB 118                                             

score      DEFB 28,28,28,28,0                                   

           DEFB "H"+x,"U"+x,"M"+x,"P"+x,"T"+x,"Y"+x,128         

           DEFB "D"+x,"U"+x,"M"+x,"P"+x,"T"+x,"Y"+x,150         

           DEFB "E"+x,"G"+x,"G"+x,128,"D"+x,"R"+x,"O"+x,"P"+x   

           DEFB 0                                               

hiscore    DEFB 28,28,28,28                                     

           DEFB 118                                             

                                                                

scrudg     EQU  $                 ; Hires Humpty screen

                                                                

vars       DEFB 128                                             

?                                                               

last       EQU  $                                               


